


Dror Bar - Natan: Classes: 2007-08: Geometry and Topology:


Shadows of the Cantor Aerogel

```
GraphicsGrid[Partition[Table[
  ArrayPlot[Table[
    If[MemberQ[IntegerDigits[x, 3]~Union~IntegerDigits[y, 3], 1], 0, 1],
    {x, 0, 3^n-1}, {y, 0, 3^n-1}
  ]],
  {n, 1, 6}
], 3]]
```



```
CC[0] = Line[{{0, 0}, {1, 1}}];
```

```
Graphics[CC[0]]
```



```
Evolve[Line[{a_, b_}]] := Module[{v, n},
  v = b - a; n = {-v[[2]], v[[1]]};
  {
    Line[{a, a+v/3}], Line[{a+2 v/3, b}],
    Line[{a+v/3+n/3, a+2 v/3+n/3}],
    Line[{a+v/3-n/3, a+2 v/3-n/3}
  ]
};
```

```
CC[n_] := CC[n] = CC[n-1] /. l_Line => Evolve[l];
```

```
CC[1]
```

```
{Line[{{0, 0}, {1/3, 1/3}}], Line[{{2/3, 2/3}, {1, 1}}],
  Line[{{0, 2/3}, {1/3, 1}}], Line[{{2/3, 0}, {1, 1/3}}]}
```

```
Graphics[CC[3]]
```

```
/// ///    /// ///
/// ///    /// ///
/// ///    /// ///
```

```
/// ///    /// ///
/// ///    /// ///
/// ///    /// ///
```

```
v = {0.6, 0.8};
```

```
basepoints = Sort[Flatten[CC[3] /. Line[{a_, _}] => a.v]]
```

```
{0., 0.0444444, 0.0592593, 0.103704, 0.133333, 0.177778, 0.177778, 0.192593, 0.222222,
 0.237037, 0.237037, 0.281481, 0.311111, 0.355556, 0.37037, 0.4, 0.414815, 0.444444,
 0.459259, 0.503704, 0.533333, 0.533333, 0.577778, 0.577778, 0.577778, 0.592593, 0.592593,
 0.622222, 0.637037, 0.637037, 0.637037, 0.666667, 0.681481, 0.711111, 0.711111, 0.711111,
 0.725926, 0.755556, 0.755556, 0.77037, 0.77037, 0.77037, 0.814815, 0.814815, 0.844444,
 0.888889, 0.903704, 0.933333, 0.948148, 0.977778, 0.992593, 1.03704, 1.06667, 1.11111,
 1.11111, 1.12593, 1.15556, 1.17037, 1.17037, 1.21481, 1.24444, 1.28889, 1.3037, 1.34815}
```

```
Measure[b_, l_] := l + Sum[
  Min[l, b[[i]] - b[[i - 1]]],
  {i, 2, Length[b]}
];
```

```
Measure[basepoints, v.{1, 1}/3^3]
```

```
1.4
```

```
CCShadow[n_, t_] := Module[{v, basepoints},
  v = {Cos[t], Sin[t]};
  basepoints = Sort[Flatten[CC[n] /. Line[{a_, _}] => a.v]];
  Measure[basepoints, v.{1, 1}/3^n]
];
```

```
{CCShadow[3, 0], CCShadow[3, Pi/4]}
```

```
{ $\frac{8}{27}$ ,  $\sqrt{2}$ }
```

```
Plot[{CCShadow[1, t], CCShadow[2, t], CCShadow[3, t], CCShadow[4, t], CCShadow[5, t]},  
  {t, 0, Pi/2}, PlotRange -> {0, Sqrt[2]}]
```

